# $Id$
# Maintainer: Jan Alexander Steffens (heftig) <jan.steffens@gmail.com>
# Contributor: Daniel J Griffiths <ghost1227@archlinux.us>
# Contributor: Corrado Primier <bardo@aur.archlinux.org>
# Contributor: William Rea <sillywilly@gmail.com>

pkgname=pavucontrol
pkgdesc="PulseAudio Volume Control"
pkgver=3.0
pkgrel=4
arch=(i686 x86_64)
url="http://freedesktop.org/software/pulseaudio/pavucontrol/"
license=(GPL)
depends=(libcanberra-pulse gtkmm3 libsigc++)
makedepends=(intltool lynx git)
optdepends=("pulseaudio: Audio backend")
_commit=9b307dcfe2051caea7145c8ab48953bb4f57005c
source=("git://anongit.freedesktop.org/pulseaudio/pavucontrol#commit=$_commit")
sha256sums=('SKIP')

prepare() {
  cd $pkgname
  NOCONFIGURE=1 ./bootstrap.sh
}

build() {
  cd $pkgname
  CXXFLAGS+=' -std=c++11'
  ./configure --prefix=/usr --enable-gtk3
  make
}

package() {
  cd $pkgname
  make DESTDIR="$pkgdir" install
}
